
1. Overview

This document lists a TTCN change needed to correct issues in the ATS ‘iwd-EUTRA-B2013-03_D13wk15’.

3. Corrections

Change 1 – Correction to local function ‘fl_GetAPNForIMS’
	Function name
	fl_GetAPNForIMS()

	Reason for change
	Implementation of endoding of APN is not as per TS 23.003 section 9.1 which states:
 
“The encoding of the APN shall follow the Name Syntax defined in RFC 2181 [18], RFC 1035 [19] and RFC 1123 [20]. The APN consists of one or more labels. 

Each label is coded as a one octet length field followed by that number of octets coded as 8 bit ASCII characters. Following RFC 1035 [19] the labels shall consist only of the alphabetic characters (A-Z and a-z), digits (0-9) and the hyphen (-). Following RFC 1123 [20], the label shall begin and end with either an alphabetic character or a digit. The case of alphabetic characters is not significant. The APN is not terminated by a length byte of zero.“

	Summary of change
	Encoding of APN Implemented acc. to TS 23.003 section 9.1.

	TTCN module
	EUTRA_NASSteps.ttcn

	MCC160 Comment
	Accepted in principle (implemented as shown below)


Before change

  //----------------------------------------------------------------------------
  /*
   * @desc      Creates the APN Name to be used if UE supports IMS, as defined in 36.508, Table 4.7.3-6
   * @param     p_CellId   to get the PLMN Id of the cell
   * @return    octetstring
   * @status    APPROVED
   */
  function fl_GetAPNForIMS(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return octetstring
  {
    var GutiParameters_Type v_Guti_Params := f_EUTRA_CellInfo_GetGuti (p_CellId);
    var PLMN_Identity v_PLMN := v_Guti_Params.PLMN_Identity;
    var hexstring v_MCC := '000'H; // initialised with 3 digits which will be overwritten
    var hexstring v_MNC := '000'H; // initialised with 3 digits which will be overwritten
    var octetstring v_APN_Name, v_Label;
    var MCC_MNC_Digit v_Digit;     // MCC-MNC-Digit ::= INTEGER (0..9)
    var integer v_Incr := 0;
    var integer I;
    if (lengthof(v_MNC) == 2) { // according to 23.003 if MNC is only 2 digits, a 0 is added on the left handside (i.e. first)
      v_Incr := 1;
    }
    for (I := 0; (I < lengthof(v_PLMN.mcc)) and (I < lengthof(v_MCC)); I := I + 1) {
      /* v_PLMN.mcc consists of 3 digits; nevertheless to be sure v_MCC is checked as well */
      v_Digit := v_PLMN.mcc[I];
      v_MCC[I] := int2hex(v_Digit, 1);
    }
    for (I := 0; (I < lengthof(v_PLMN.mnc)) and (I < lengthof(v_MNC)); I := I + 1) {
      /* v_PLMN.mcc consists of 2 or 3 digits; nevertheless to be sure v_MNC is checked as well */
      v_Digit := v_PLMN.mnc[I];
      v_MNC[I+v_Incr] := int2hex(v_Digit, 1);
    }
    // 23.003 describes APN as a concatenation of labels, each with a single octet length field first
    v_Label := char2oct ("IMS");
    v_APN_Name := int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("apn");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("epc");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("mnc") & hex2oct(v_MNC);
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("mcc") & hex2oct(v_MCC);
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("3gppnetwork");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("org");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    return v_APN_Name;
  }
After change

  //----------------------------------------------------------------------------
  /*
   * @desc      Creates the APN Name to be used if UE supports IMS, as defined in 36.508, Table 4.7.3-6
   * @param     p_CellId   to get the PLMN Id of the cell
   * @return    octetstring
   * @status    APPROVED
   */
    function fl_GetAPNForIMS(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return octetstring
  {
    // BZ 43545
    var GutiParameters_Type v_Guti_Params := f_EUTRA_CellInfo_GetGuti (p_CellId);
    var PLMN_Identity v_PLMN := v_Guti_Params.PLMN_Identity;
    var octetstring v_MCC;
    var octetstring v_MNC;
    var octetstring v_APN_Name, v_Label;
    var MCC_MNC_Digit v_Digit;     // MCC-MNC-Digit ::= INTEGER (0..9)
    var integer I;
    for (I := 0; I < lengthof(v_PLMN.mcc); I := I + 1) {
      /* v_PLMN.mcc consists of 3 digits; nevertheless to be sure v_MCC is checked as well */
      v_Digit := v_PLMN.mcc[I];
      if (I == 0) {
        v_MCC := int2oct(v_Digit + 48, 1); // Convert digit + 0x30
      } else {
        v_MCC := v_MCC & int2oct(v_Digit + 48, 1); // Convert digit + 0x30
      }
    }
    for (I := 0; I < lengthof(v_PLMN.mnc); I := I + 1) {
      v_Digit := v_PLMN.mnc[I];
       if (I == 0) {
         v_MNC := int2oct(v_Digit + 48, 1); // Convert digit + 0x30
       } else {
         v_MNC := v_MNC & int2oct(v_Digit + 48, 1); // Convert digit + 0x30
       }
    }
    /*  TS36.508 Table 4.7.3-6: ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST specifies APN as
          IMS.apn.epc.mnc<MNC>.mcc<MCC>.3gppnetwork.org
          The <MNC> and <MCC> are set to the same values as in IMSI.
    TS 24.301 9.9.4.1 Access point name
       -> See  in 3GPP TS 24.008, clause 10.5.6.1
         -> See 23.003 clause 9.1.
          The encoding of the APN shall follow the Name Syntax defined in RFC 2181 [18], 
          RFC 1035 [19] and RFC 1123 [20]. 
          The APN consists of one or more labels. 
          Each label is coded as a one octet length field followed by that number of octets coded as 8 bit ASCII characters. Following RFC 1035 [19] the labels shall consist only of the alphabetic characters (A-Z and a-z), digits (0-9) and the hyphen (-). Following RFC 1123 [20], the label shall begin and end with either an alphabetic character or a digit. The case of alphabetic characters is not significant. The APN is not terminated by a length byte of zero.    */
    v_Label := char2oct ("IMS");
    v_APN_Name := int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("apn");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("epc");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    if (lengthof(v_MNC) == 2) {
      v_Label := char2oct ("mnc") & '30'O & v_MNC; // Add "0" (0x30) , See 23.003 clause 9.1.2.
    } else {
      v_Label := char2oct ("mnc") & v_MNC; 
    }
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("mcc") & v_MCC;
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("3gppnetwork");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    v_Label := char2oct ("org");
    v_APN_Name := v_APN_Name & int2oct(lengthof(v_Label), 1) & v_Label;
    return v_APN_Name;
  }
MCC160 Implementation
	  function f_APN_MccMncLabel(charstring p_MncMcc,

                             charstring p_Digits) return charstring

  { /* in case of the MCC there are 3 digits; for MNC it may be 2 or 3 digits */

    var integer v_DigitCnt := lengthof(p_Digits);

    var charstring v_Label := "";

    select (v_DigitCnt) {

      case (2) { v_Label := p_MncMcc & "0" & p_Digits; }

      case (3) { v_Label := p_MncMcc & p_Digits; }

      case else { FatalError(__FILE__, __LINE__, ""); }

    }

    return v_Label;

  }

	  function f_APN_Encode(CharStringList_Type p_LabelList) return octetstring

  {

    var octetstring v_EncodedAPN := ''O;

    var octetstring v_LabelOctets;

    var integer v_Length;

    var integer i;

    for (i := 0; i < lengthof(p_LabelList); i := i + 1) {

      v_LabelOctets := char2oct(p_LabelList[i]);

      v_Length := lengthof(v_LabelOctets);

      v_EncodedAPN := v_EncodedAPN & int2oct(v_Length, 1) & v_LabelOctets;

    }

    return v_EncodedAPN;

  }

	  function f_EUTRA_GetAPNForIMS(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return octetstring

  {

    var GutiParameters_Type v_Guti_Params := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var PLMN_Identity v_PLMN := v_Guti_Params.PLMN_Identity;

    var charstring v_MNC := "";

    var charstring v_MCC := "";

    var CharStringList_Type v_LabelList;

    var integer i;

    for (i := 0; i < lengthof(v_PLMN.mnc); i := i + 1) { v_MNC := v_MNC & int2str(v_PLMN.mnc[i]); }

    for (i := 0; i < lengthof(v_PLMN.mcc); i := i + 1) { v_MCC := v_MCC & int2str(v_PLMN.mnc[i]); }

    v_LabelList := {"IMS", "apn", "epc", f_APN_MccMncLabel("mnc", v_MNC), f_APN_MccMncLabel("mcc", v_MCC), "3gppnetwork", "org" };

    return f_APN_Encode(v_LabelList);

  }

